Corrigé du TP : polynômes multivariés et combinatoire

O. Marguin - 18 mai 2011

1. Nombres porte-bonheur

```
> restart:
```

Si N est un entier porte-bonheur à n chiffres, on a deux cas (disjoints) :
1) ou bien N a au moins un 13 parmi ses $n-1$ premiers chiffres ; le dernier chiffre peut être quelconque (de 0 à 9) : il y a $10\,u_{n-1}$ nombres de ce type,
2) ou bien N n'a pas de 13 parmi ses $n-1$ premiers chiffres ; alors N se termine par 13, et 13 n'apparaît pas dans les $n-2$ premiers chiffres : il y a $10^{n-2}-u_{n-2}$ nombres de ce type.
D'où la formule de récurrence :

```
> eq:=u(n)=10*u(n-1)+10^(n-2)-u(n-2);
```

$$eq := u(n) = 10\,u(n-1) + 10^{n-2} - u(n-2)$$

```
> # résolution de la récurrence :
> s:=rsolve({eq,u(0)=0,u(1)=0},u(n));
```

$$s := \frac{1}{24}\,\frac{\sqrt{6}\left(-\frac{1}{-5+2\sqrt{6}}\right)^n}{-5+2\sqrt{6}} - \frac{1}{24}\,\frac{\sqrt{6}\left(-\frac{1}{-5-2\sqrt{6}}\right)^n}{-5-2\sqrt{6}} + 10^n$$

```
> # calcul de u(20):
> u20:=factor(subs(n=20,s));
```

$$u20 := 17536209731411055499$$

```
> # détermination de la série génératrice de u(n):
> uinitiales:=[seq(factor(subs(n=k,s)),k=0..20)];
```

$uinitiales$:= [0, 0, 1, 20, 299, 3970, 49401, 590040, 6850999, 77919950, 872348501, 9645565060, 105583302099, 1146187455930, 12356291257201, 132416725116080, 1411810959903599, 14985692873919910, 158445117779295501, 1669465484919035100, 17536209731411055499]

```
> with(gfun):
> fg:=guessgf(uinitiales,X);
```

$$fg := \left[\frac{X^2}{1-20\,X+101\,X^2-10\,X^3},\ ogf\right]$$

```
> S:=fg[1];
```

$$S := \frac{X^2}{1-20\,X+101\,X^2-10\,X^3}$$

```
> # vérification:
> coeff(series(S,X=0,21),X,20)-u20;
```

$$0$$

```
>
```

2. Problème de vases

```
> restart:
> # question 1 :
> transvase:=proc(M)
  # envisage tous les transvasements possibles à partir de la
  configuration M
  # et renvoie l'ensemble des configurations obtenues sous
  forme d'un polynôme
     global capacites,n;
     local i,j,e1,e2,verse,S;
     S:=0;
     for i in {$1..n} do
        for j in {$1..n} minus {i} do
           # verser i dans j si c'est possible :
           e1:=degree(M,X[i]);
           e2:=degree(M,X[j]);
           verse:=min(capacites[j]-e2,e1);
           if verse<>0 then
              S:=S+M*(X[j]/X[i])^verse;
           end if
        end do
     end do;
     return S
  end proc:
> # question 2 :
> configs:=proc(M,k)
  # totalise toutes les configurations obtenues à partir de M
  après k transvasements
     local P,i;
     P:=transvase(M);
     for i to k-1 do P:=map(transvase,P) end do;
     return P
  end proc:
> # application numérique :
> n:=3;capacites:=[50,33,19];Minit:=X[1]^capacites[1];
```

$$n := 3$$

$$capacites := [50, 33, 19]$$

$$Minit := X_1^{50}$$

```
> P:=configs(Minit,5);
```

$$P := 9\,X_3^{17}X_1^{33} + 5\,X_1^{31}X_2^{19} + 22\,X_3^{19}X_2^{31} + 4\,X_1^{17}X_2^{14}X_3^{19} + 14\,X_2^{33}X_3^{17} + 20\,X_1^{50} + 37\,X_1^{17}X_2^{33} + 36\,X_1^{31}X_3^{19} + 7\,X_1^{36}X_2^{14} + 9\,X_2^{31}X_1^{19} + 7\,X_1^{12}X_2^{19}X_3^{19} + X_1^{33}X_2^{17} + 2\,X_2^{12}X_1^{19}X_3^{19} + X_1^{14}X_2^{17}X_3^{19} + X_1^{3}X_3^{14}X_2^{33} + X_2^{12}X_1^{38} + X_1^{45}X_3^{5}$$

```
> nops(P);
```

$$17$$

Donc après 5 transvasements, il y a 17 configurations, dont les probabilités sont données par les coefficients du polynôme :

```
> P/convert([coeffs(P)],`+`);
```

$$\frac{3}{59}X_3^{17}X_1^{33} + \frac{5}{177}X_1^{31}X_2^{19} + \frac{22}{177}X_3^{19}X_2^{31} + \frac{4}{177}X_1^{17}X_2^{14}X_3^{19} + \frac{14}{177}X_2^{33}X_3^{17} + \frac{20}{177}X_1^{50} + \frac{37}{177}X_1^{17}X_2^{33} + \frac{12}{59}X_1^{31}X_3^{19} + \frac{7}{177}X_1^{36}X_2^{14} + \frac{3}{59}X_2^{31}X_1^{19} + \frac{7}{177}X_1^{12}X_2^{19}X_3^{19} + \frac{1}{177}X_1^{33}X_2^{17} + \frac{2}{177}X_2^{12}X_1^{19}X_3^{19} + \frac{1}{177}X_1^{14}X_2^{17}X_3^{19} + \frac{1}{177}X_1^{3}X_3^{14}X_2^{33} + \frac{1}{177}X_2^{12}X_1^{38} + \frac{1}{177}X_1^{45}X_3^{5}$$

```
> # question 3
> bon:=proc(M) # teste si la configuration M est bonne : l'un
  des exposants est égal à mesure
      global mesure,n;
      member(mesure,{seq(degree(M,X[i]),i=1..n)})
  end proc:
```

```
> cherche:=proc(M,P,nbcoups)
  # cherche une solution en au plus nbcoups à partir de la
  configuration M
  # P est un polynôme décrivant la suite des configurations
  menant à M
     global solution;
     local Q,R,MM;
     if nbcoups>0 then
       Q:=Y*P+M;  # Y est une indéterminée supplémentaire
       R:=transvase(M);
       for MM in {op(R)} do
          if bon(MM) then
             solution:=Y*Q+MM
          else if not has(Q,MM) then  # pour éviter les
  cycles
                 cherche(MM,Q,nbcoups-1)
              end if
          end if
       end do
     end if
  end proc:
```

(remarque : les puissances de l'indéterminée supplémentaire Y indiqueront l'ordre des coups).

```
> # application numérique :
> mesure:=1;
```

$$mesure := 1$$

```
> solution:=0: cherche(Minit,0,19): solution;
```

$$0$$

Donc pas de solution en 19 coups ou moins. En revanche, il y en a une en 20 coups :

```
> solution:=0: cherche(Minit,0,20): subs(Y=1,sort(expand
  (solution),Y));
```

$$X_1^{50} + X_1^{31} X_3^{19} + X_1^{31} X_2^{19} + X_1^{12} X_2^{19} X_3^{19} + X_1^{12} X_2^{33} X_3^{5} + X_1^{45} X_3^{5} + X_1^{45} X_2^{5} + X_1^{26} X_2^{5} X_3^{19} + X_1^{26} X_2^{24} + X_1^{7} X_2^{24} X_3^{19} + X_1^{7} X_2^{33} X_3^{10} + X_1^{40} X_3^{10} + X_1^{40} X_2^{10} + X_1^{21} X_2^{10} X_3^{19} + X_1^{21} X_2^{29} + X_1^{2} X_2^{29} X_3^{19} + X_1^{2} X_2^{33} X_3^{15} + X_1^{35} X_3^{15} + X_1^{35} X_2^{15} + X_1^{16} X_2^{15} X_3^{19} + X_1^{16} X_2^{33} X_3$$

polynôme qui décrit la suite des configurations menant au résultat.

```
>
```